

## 7 The Integral

In previous chapters you have studied how to find derivatives and have seen how they have many useful interpretations. Before Chapter 4 only the basic limit definition for computing derivatives was available, but in Chapter 4 you studied how to differentiate many different functions exactly either by using Maple V or by hand. Up until now most integrals that have been calculated have only been approximated by using limits of left-hand or right-hand sums. In this chapter you will learn how to calculate the definite integral of certain functions exactly, by using the Fundamental Theorem of Calculus. More efficient numerical methods will also be introduced.

### 7.1 Some Basic Formulas

Maple V is an excellent tool for computing antiderivatives. In this section Maple V will be used to establish some basic formulas. Remember the problem in finding an antiderivative for  $f(x)$  is to find a function  $F(x)$  such that  $F'(x) = f(x)$ . Thus one can always determine the correctness of the antiderivative by differentiation. For example the Maple V segment

```
> Int(x^n,x): " = value(") + C;
```

$$\int x^n dx = \frac{x^{n+1}}{n+1} + C$$

illustrates that if  $n \neq -1$  then the general antiderivative or the *indefinite integral* of  $x^n$  is  $\frac{x^{n+1}}{n+1} + C$ . This can be checked by differentiation.

```
> diff(rhs("),x);
```

$$\frac{x^{n+1}}{x}$$

```
> simplify(");
```

$$x^n$$

The following is a basic integration formula.

$$\int x^n dx = \frac{x^{n+1}}{n+1} + C, \quad n \neq -1$$

What happens in the case that  $n = -1$ ? What is the antiderivative of  $\frac{1}{x}$ ?

```
> Int(subs(n=-1,x^n),x): " = value(")+C;
```

$$\int x^{-1} dx = \ln(x) + C$$

This implies that an antiderivative of  $\frac{1}{x}$  is  $\ln x$ . Recall that the domain of  $\ln x$  is  $\{x|x > 0\}$  and hence the above formula is undefined if  $x \leq 0$ . The Fundamental Theorem of Calculus holds only for continuous functions and  $\frac{1}{x}$  is discontinuous at  $x = 0$ . This means that one shouldn't expect an antiderivative for  $\frac{1}{x}$  to be defined at  $x = 0$ . This function is continuous over every interval that does not include  $x = 0$ . Thus one can expect the function defined by  $f(x) = 1/x$ , for  $x < 0$  to have an antiderivative. If  $x < 0$  then  $\ln(-x)$  makes sense. Furthermore, by the Chain Rule we have

$$\frac{d}{dx} \ln(-x) = (-1) \frac{1}{-x} = \frac{1}{x}.$$

Therefore an antiderivative for  $f(x)$  is  $\ln(-x) = \ln|x|$ . The formula for the indefinite integral of  $\frac{1}{x}$  is given in the box below.

$$\int \frac{1}{x} dx = \ln|x| + C.$$

Recall that the exponential function is equal to its derivative and so the following is true.

$$\int e^x dx = e^x + C.$$

For the sin and cos functions we have

> Int(sin(x), x): "=value(")+C;  

$$\int \sin(x) dx = -\cos(x) + C$$

> Int(cos(x), x): "=value(")+C;  

$$\int \cos(x) dx = \sin(x) + C$$

$$\int \sin(x) dx = -\cos(x) + C$$

$$\int \cos(x) dx = \sin(x) + C$$

In practice a student should be able to do a certain amount of differentiation and integration by hand even when he or she has a computer algebra system such as Maple V. This is analogous to the fact that every one should know how to perform simple arithmetic computations even when hand calculators are available, or one needs to know definitions of some words even when a dictionary is available. Every calculus student should know the preceding formulas (and some others) for finding indefinite integrals, but calculus students who have the use of computer algebra systems, like Maple V, need not spend as much time memorizing integration formulas as students in the past had to do.

The following is the indefinite integral version of the facts given in Chapter 6 about adding two definite integrals and multiplying an integral by a constant. You will find it very useful for finding antiderivatives of linear combinations of functions with known antiderivatives.

**Facts about Sums and Constant Multiples**

$$\int (f(x) + g(x)) dx = \int f(x) dx + \int g(x) dx$$

$$\int cf(x) dx = c \int f(x) dx.$$

**Example 7.1.1** Evaluate the following definite integrals:

$$\int_1^2 (1/x + 1/x^2) dx, \quad \int_0^1 37x^5 dx, \quad \int_0^\pi (3e^x - 5 \cos x) dx.$$

**Solution:**

$$\int_1^2 (1/x + 1/x^2) dx = \ln x - \frac{1}{x} \Big|_1^2 = \ln 2 - 1/2 + 1 = \ln 2 + 1/2.$$

$$\int_0^1 37x^5 dx = 37 \frac{x^6}{6} \Big|_0^1 = 37/6.$$

$$\int_0^{\pi} (3e^x - 5 \cos x) dx = 3e^x - 5 \sin x \Big|_0^{\pi} = 3e^{\pi} - 3.$$

Each of the preceding examples was simple enough that it was easily worked by hand. Sometimes one encounters more complicated functions and Maple V is helpful in such situations.

**Example 7.1.2** Evaluate the following integrals:

$$\int \cos \sqrt{x} dx, \quad \int \cos x^2 dx, \quad \int_0^1 \cos x^2 dx.$$

**Solution:**

$$\begin{aligned} &> \text{Int}(\cos(\text{sqrt}(x)), x) : "=value(")+C; \\ &\int \cos(\sqrt{x}) dx = 2 \cos(\sqrt{x}) + 2 \sqrt{x} \sin(\sqrt{x}) + C \end{aligned}$$

$$\begin{aligned} &> \text{diff}(\text{rhs}("), x); \\ &\cos(\sqrt{x}) \end{aligned}$$

$$\begin{aligned} &> \text{Int}(\cos(x^2), x) : "= value(")+C; \\ &\int \cos(x^2) dx = \sqrt{2} \sqrt{\pi} \text{FresnelC}\left(\frac{\sqrt{2}x}{\sqrt{\pi}}\right) 1/2 + C \end{aligned}$$

You will not be expected to know the properties of the function called *FresnelC* in this course.

$$\begin{aligned} &> \text{diff}(\text{rhs}("), x); \\ &\cos(x^2) \end{aligned}$$

$$\begin{aligned} &> \text{Int}(\cos(x^2), x=0..1) : "= value("); \\ &\int_0^1 \cos(x^2) dx = \frac{\sqrt{2} \sqrt{\pi} \text{FresnelC}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\right)}{2} \end{aligned}$$

$$\begin{aligned} &> \text{Int}(\cos(x^2), x=0..1) = \text{evalf}(\text{rhs}(")); \\ &\int_0^1 \cos(x^2) dx = 0.9045242375 \end{aligned}$$

### Exercises 7.1

1. Use Maple V to evaluate the following:

(a)

$$\int \exp(3x) \sin(5x) dx$$

(b)

$$\int \frac{\ln x}{x^4} dx$$

(c)

$$\int \sin 3x \cos^2 2x dx$$

(d)

$$\int_0^2 \sqrt{4-x^2} dx$$

2. (a) Use Maple V to find

$$\int \sec^2 x dx.$$

(b) Based on your answer in (a) what should the derivative of  $\tan x$  be?

(c) Use Maple V to find the derivative of  $\tan x$ . Explain any apparent discrepancies.

3. Enter the following Maple V command in a worksheet.

```
> seq(int(sin(n*x)/sin(x), x=0..Pi), n=1..10);
```

Use your results to make a conjecture about the value of

$$\int_0^\pi \frac{\sin nx}{\sin x} dx$$

for positive integers  $n$ .

## 7.2 Finding Integrals by the Method of Substitution

Remember that the only thing you are required to do when finding an antiderivative of a given function  $f(x)$  is to find a function  $F(x)$  such that  $F'(x) = f(x)$ . Over the years there have been many techniques developed for calculating integrals. The approach illustrated in this section can be regarded as a method of reversing the Chain Rule which was given in Chapter 4. Recall that if  $f$  and  $g$  are given and the composition  $f \circ g$  is defined then the Chain Rule says

$$(f \circ g)'(x) = f'(g(x)) \cdot g'(x).$$

The indefinite integral that corresponds to the last formula is

$$\int f'(g(x)) \cdot g'(x) dx = (f \circ g)'(x) + C.$$

To illustrate this, consider the problem of integrating the function  $h(x) = 2x \cos x^2$ . Is there a function whose derivative is equal to  $h(x)$ ? By applying the Chain Rule it can be seen that

$$\frac{d}{dx} \sin x^2 = 2x \cos x^2 = h(x).$$

It follows that

$$\int h(x) dx = \int 2x \cos x^2 dx = \sin x^2 + C.$$

Another way to think of this problem is to look at the integral

$$\int 2x \cos x^2 dx$$

and recognize that the integrand looks like the result of applying the Chain Rule in the situation where  $f(u) = \sin(u)$ , and  $g(x) = x^2$ . A device that helps us in this is to think of this process as changing variables. Thus we think of making the substitution  $u = g(x) = x^2$  into the integral  $\int 2x \sin x^2 dx$ . This is written

$$\int 2x \cos x^2 dx = \int \cos x^2 (2x dx) = \int \cos u du = \sin u + C = \sin x^2 + C.$$

In order that the method of substitution be successful for a given function  $h$  it must be recognized that there are two functions  $f$  and  $g$  such that  $h(x) = (f \circ g)'(x) = f'(g(x)) \cdot g'(x)$ .

Observe that  $\int \cos x^2 dx$  does not fall into this category, since if we try the obvious  $\sin x^2$  as an antiderivative then its derivative fails to equal  $\cos x^2$  because of the chain rule.

**Example 7.2.1** Find  $\int x^4 \sqrt{x^5 + 3} dx$ .

**Solution:** This problem suggests the Chain Rule with  $f(u) = \sqrt{u}$  and  $u = x^5 + 3$ . Lets see what happens to the integral if we make this substitution. Since  $du = 5x^4 dx$  or  $x^4 dx = \frac{1}{5} du$  in this case we have

$$\int x^4 \sqrt{x^5 + 3} dx = \int \sqrt{x^5 + 3} (x^4 dx) = \int \sqrt{u} \frac{du}{5} = \frac{1}{5} \frac{u^{3/2}}{3/2} + C = \frac{2}{15} (x^5 + 3)^{3/2} + C.$$

Check this result by differentiation

$$\frac{d}{dx} \left( \frac{2}{15} (x^5 + 3)^{3/2} + C \right) = \frac{2}{15} \cdot \frac{3}{2} (x^5 + 3)^{1/2} \cdot (5x^4) = x^4 \sqrt{x^5 + 3}$$

In the previous example one is able to find a substitution that transforms the integrand into one that can be readily integrated. Do you think that the method of substitution can be applied to the following integrals?

$$\int \sqrt{x^5 + 3} dx, \quad \int x^2 \sqrt{x^5 + 3} dx$$

**Example 7.2.2** Evaluate  $\int x^3 e^{x^4+4} dx$ .

**Solution:** In this case we set  $u = x^4 + 4$  and  $du = 4x^3 dx$  or  $x^3 dx = \frac{du}{4}$ . This leads to the following calculation.

$$\int x^3 e^{x^4+4} dx = \int e^{x^4+4} (x^3 dx) = \int e^u \frac{du}{4} = \frac{1}{4} e^u + C = \frac{1}{4} e^{x^4+4} + C.$$

Now check the result by differentiation.

$$\frac{d}{dx} \left( \frac{1}{4} e^{x^4+4} + C \right) = \frac{1}{4} e^{x^4+4} \cdot (4 \cdot x^3) = x^3 e^{x^4+4}.$$

When using Maple V one can solve both of the above problems very easily.

```
> Int(x^4*sqrt(x^5+1),x): " = value(") + C;
```

$$\int x^4 \sqrt{x^5 + 1} dx = \frac{2}{15} (x^5 + 1)^{3/2} + C$$

```
> Int(x^3*exp(x^4+1),x) : " = value(") + C;
```

$$\int x^3 e^{x^4+1} dx = \frac{e^{x^4+1}}{4} + C$$

One might ask the question: why study the substitution method for integrating complicated expressions by hand when the problem can be solved so effortlessly by Maple V? It's true that in practice, when one has access to computer algebra systems like Maple V, one can quickly integrate many complicated problems without needing to know a large bag of tricks. Nevertheless, in order to build up an intuition for the manipulation of functions and to develop skills in dealing with them, we will now introduce a Maple V procedure, **changevar** that is a Maple V tool for integrating functions by the method of substitution. The syntax is **changevar**( $s, f, u$ ), where  $s$  is an expression of the form  $h(x) = g(u)$ , defining  $x$  as a function of  $u$ ;  $f$  is an expression such as  $\text{Int}(F(x), x = a \dots b)$ ; and  $u$  is the name of the new integration variable. The procedure **changevar** is part of the **student** package and hence requires either the **with(student)** command or is invoked using the long version **student[changevar]**.

To illustrate this procedure the first example above will now be reworked using **changevar**. Invoke the **student** package.

```
> with(student):
```

Define the integral.

```
> I1 := Int(x^4*sqrt(x^5+1), x, u);
```

$$I1 := \int x^4 \sqrt{x^5 + 1} dx$$

When this expression was integrated by hand the substitution  $u = x^5 + 1$  was used. The proper syntax in this case is  $x^5 + 1 = u$ . This reduces the integration problem the following:

```
> I2 := changevar(x^5+1=u, I1);
```

$$I2 := \int \frac{\sqrt{u}}{5} du$$

The new integration problem is equivalent to the original one. However, the new problem is simpler and can be integrated by the power rule discussed in the previous section.

```
> value(");
```

$$\frac{2u^{3/2}}{15}$$

The value of the original integral is now obtained by using **subs** with  $u = x^5 + 1$ .

```
> subs(u = x^5+1, ");
```

$$\frac{2(x^5 + 1)^{3/2}}{15}$$

As with any antidifferentiation problem you can check your result by differentiation.

```
> diff(", x);
```

$$x^4 \sqrt{x^5 + 1}$$

When evaluating a definite integral it is usually easier to change the limits of integration defined by the transformation.

**Example 7.2.3** Evaluate the definite integral  $\int_0^1 x^4 \sqrt{x^5 + 3} dx$ .

**Solution:** Enter the following Maple V statement.

```
> II1 := Int(x^4*sqrt(x^5+1), x=0..1);
```

$$II1 := \int_0^1 x^4 \sqrt{x^5 + 1} dx$$

Proceed just as you would when evaluating an indefinite integral with **changevar**

```
> II2 := student[changevar](x^5+1=u, II1, u);
```

$$II2 := \int_1^2 \frac{\sqrt{u}}{5} du$$

Observe the new limits of integration. Corresponding to  $x = 0$  for the original lower limit is

$$u = 0^5 + 1 = 1,$$

and corresponding to  $x = 1$  in the original upper limit is  $u = 1^5 + 1$ . One can now evaluate the transformed integral without the need for substituting in the original variables.

```
> value(");
```

$$\frac{4\sqrt{2}}{15} - 2/15$$

Therefore,

$$\int_0^1 x^4 \sqrt{x^5 + 3} dx = \frac{4\sqrt{2}}{15} - 2/15.$$

When using the method of substitution either by hand or with the **changevar** command the goal is to reformulate an integral into a form in which the integral follows from a basic formula.

**Example 7.2.4** Evaluate  $\int x \frac{1}{\sqrt{4-9x^2}} dx$ .

**Solution:** With some practice you should eventually be able to solve this by hand. The integral will be calculating by using **changevar**.

```
> I3 := Int(x/sqrt(4-9*x^2), x);
```

$$I3 := \int \frac{x}{\sqrt{4-9x^2}} dx$$

What substitution should be used? In this case if  $u = 4 - 9x^2$ , then  $du = -18x dx$ . This appears to be worth trying.

```
> I4 := student[changevar](4-9*x^2=u, I3, u);
```

$$I4 := \int -\frac{1}{18\sqrt{u}} du$$

This integral appears to be much simpler. Nevertheless let's simplify it further.

```
> I5 := simplify(I4);
```

$$I5 := -\frac{\int \frac{1}{\sqrt{u}} du}{18}$$

The goal here is to reduce the original problem to one of the basic formulas. We have done it and now its okay to apply **value**.

```
> I6 := value(I5);
```

$$I6 := -\frac{\sqrt{u}}{9}$$

Returning to the original variables one has

```
> I7 := subs(u=4-9*x^2, I6);
```

$$I7 := -\frac{\sqrt{4-9x^2}}{9}$$

A check of this is performed by differentiation:

```
> diff(I7, x);
```

$$\frac{x}{\sqrt{4-9x^2}}$$

It follows that

$$\int x \frac{1}{\sqrt{4-9x^2}} dx = -\frac{\sqrt{4-9x^2}}{9} + C.$$

Remember that you should regard **changevar** as a tool to help you learn the method of substitution. In practical situations you should be able to evaluate easy integrals by hand and compute complicated integrals via Maple V by using the **int** procedure.

## Exercises 7.2

1. Evaluate each of the following using the method of substitution and the check your answers using Maple V and **changevar**.

(a)  $\int 5\sqrt{2+3x} \, dx$

(b)  $\int 7x^3 \sin(x^4 + 2) \, dx$

(c)  $\int_0^{1/2} x \frac{1}{\sqrt{1-4x^2}} \, dx$

(d)  $\int \cos(3x+2) \, dx$

(e)  $\int \sin^5(2x) \cos(2x) \, dx$ , try  $u = \sin(5x)$

2. If  $f$  is integrable then

$$\int_0^{\pi/2} f(\cos x) \, dx = \int_0^{\pi/2} f(\sin x) \, dx = \frac{1}{2} \int_0^{\pi} f(\sin x) \, dx;$$

$$\int_0^{m\pi} f(\cos^2 x) \, dx = m \int_0^{\pi} f(\cos^2 x) \, dx.$$

3. Show that if  $m \neq n$  then,

$$\int_0^{\pi} \cos mx \cos nx \, dx = \int_0^{\pi} \sin mx \sin nx \, dx = 0,$$

but if  $m=n$ , then each integral is equal to  $\frac{\pi}{2}$ . Also,

$$\int_0^{\pi} \cos mx \sin nx = \frac{2n}{n^2 - m^2},$$

if  $n - m$  is odd, but

$$\int_0^{\pi} \cos mx \sin nx = 0,$$

if  $n - m$  is even. (Hint: Recall the identities

$$\sin(A+B) = \sin A \cos B + \cos A \sin B$$

and

$$\cos(A+B) = \cos A \cos B - \sin A \sin B.)$$

### 7.3 Integration by Parts

In the preceding section it was seen that the method of substitution for evaluating integrals is essentially a restatement of the Chain Rule for differentiation. In this section another important integration technique known as *integration by parts* will be presented. This latter method is merely a restatement of the rule for differentiating a product. Recall that if  $u(x)$  and  $v(x)$  are differentiable functions then the rule for differentiation of their product is

$$\frac{d}{dx}(uv) = u'v + uv'.$$

The rule for integration by parts is obtained by integrating both sides of the last equality

$$\int \frac{d}{dx}(uv) \, dx = \int (uv' + u'v) \, dx = \int uv' \, dx + \int u'v \, dx.$$

Observe that the first term in the last set of equations is the antiderivative of the derivative of  $uv$ , and thus it follows that

$$uv = \int uv' \, dx + \int u'v \, dx.$$

Finally, by solving the last equation for  $\int uv' \, dx$  the integration by parts formula follows.



**Integration by Parts**

Let  $u(x)$  and  $v(x)$  be differentiable functions then

$$\int u(x)v'(x) dx = u(x)v(x) - \int u'(x)v(x) dx.$$

As an illustration of how integration by parts is used consider the following integral:

$$\int xe^x dx.$$

For this integral let  $u(x) = x$ , and  $v'(x) dx = e^x dx$ . The product of these terms is the integrand in the integral on the left hand side of the integration by parts formula. For the first term on the right-hand side one must determine  $v(x)$  from the equality  $v'(x) dx = e^x dx$ . Upon integration one has  $v(x) = e^x$ . Differentiation of the equality  $u(x) = x$  gives  $u'(x) = 1$ . Substituting these results into the integration by parts formula gives

$$\int xe^x dx = xe^x - \int 1 \cdot e^x dx.$$

Integrating the integral on the right-hand side yields the formula

$$\int xe^x dx = xe^x - e^x + C.$$

In applications one usually needs to compute a definite integral. The integration by parts formula for definite integrals is given below.

**Integration by Parts for Definite Integrals**

Let  $u(x)$  and  $v(x)$  be differentiable functions then

$$\int_a^b u(x)v'(x) dx = u(x)v(x) \Big|_a^b - \int_a^b u'(x)v(x) dx.$$

For purposes of illustration calculate

$$\int_{-1}^3 xe^x dx.$$

Just as before set  $u = x$ , and  $v' = e^x$ . Using the integration by parts formula

$$\int_{-1}^3 xe^x dx = x \cdot e^x \Big|_{-1}^3 - \int_{-1}^3 e^x dx = ((3e^3 - (-1)e^{-1}) - (e^3 - e^{-1})) = 2(e^3 + \frac{1}{e}).$$

**General Principles for Applying Integration by Parts**

The reason to use the method of integration by parts is to express a complicated integral into simpler parts.

1. Make sure that you set  $v'$  equal to a function for which you are able to find an antiderivative  $v$ .
2. Make sure that  $u'$  is simpler than  $u$  (or at least no more complicated than  $u$ ).
3. Try to make sure  $v$  is simpler than  $v'$  (or at least no more complicated than  $v'$ ).

The above principles explain the reason for making the choices  $u = x$ , and  $v' dx = e^x$ . What if, instead, the choices  $u = e^x$  and  $v' dx = x$  had been made? Then  $u' = e^x$ , which is no more complicated than  $u$ , but  $v = \frac{x^2}{2}$ . The result is the following:

$$\int x e^x dx = e^x \frac{x^2}{2} - \int \frac{x^2}{2} e^x dx,$$

which is more complicated than the original problem.

**Example 7.3.1** Integrate  $\int x \ln x dx$ .

**Solution:** In this problem set  $v' dx = x dx$  (even though it seems to contradict the third principle above) then  $v = \frac{x^2}{2}$ . This is required because the first principle states that you must be able to integrate  $v'$  and if you set  $v' = \ln x$ , then you can't easily find  $v$ . Let  $u = \ln x$ , which means  $u' = \frac{1}{x}$  and then apply integration by parts

$$\int x \ln x dx = (\ln x) \frac{x^2}{2} - \int \frac{1}{x} \frac{x^2}{2} dx,$$

and hence

$$\int x \ln x dx = \frac{x^2 \ln x}{2} - \int \frac{x}{2} = \frac{x^2 \ln x}{2} - \frac{x^2}{4} + C.$$

The **student** package has a procedure called **intparts** that helps you to practice the technique of integration by parts. The syntax for **intparts** is **intparts**( $f, u$ ), where  $f$  is an expression of the form  $\text{Int}(u dv, x)$ , and  $u$  is the factor of the integrand to be differentiated. We now give a few examples.

**Example 7.3.2** Use **intpart** to integrate

$$\int x e^x dx, \quad \int_{-1}^3 x e^x dx, \quad \text{and} \quad \int x \ln x dx.$$

**Solution:** Since we let  $u = x$  for this problem

```
> I1 := Int(x*exp(x), x);
```

$$I1 := \int x e^x dx$$

```
> I2 := student[intparts](I1, x);
```

$$I2 := x e^x - \int e^x dx$$

This is mostly an instructional tool for learning the technique of integration by parts. You have worked the problem correctly only when the integrals on the right-hand side are integrable directly from one of the basic formulas. The above Maple V output passes the test so apply **value**.

```
> I2 := value(I2);
```

$$x e^x - e^x$$

It follows that  $\int x e^x dx = x e^x - e^x + C$ .

Now calculate the definite integral. Since the final formula gives an antiderivative for  $x e^x$ , the answer can be calculated in the usual manner, *i.e.* using the fundamental formula

$$\int_a^b f(x) dx = F(b) - F(a).$$

```
> subs(x=3, I2) - subs(x=-1, I2);
```

$$2e^3 + 2e^{-1}$$

Conclude that

$$\int_{-1}^3 xe^x dx = 2e^3 + 2e^{-1}.$$

You can also evaluate a definite integral directly using **intparts**.

```
> I1a := Int(x*exp(x), x=-1..3);
```

$$I1a := \int_{-1}^3 xe^x dx$$

```
> I2a := student[intparts](I1a, x);
```

$$I2a := 3e^3 + e^{-1} - \int_{-1}^3 e^x dx$$

```
> value(I2a);
```

$$2e^3 + 2e^{-1}$$

Now for the third integral.

```
> I3 := Int(x*ln(x), x);
```

$$I3 := \int x \ln(x) dx$$

One should set  $u = \ln x$ , when using integration by parts for this integral.

```
> I4 := student[intparts](I3, ln(x));
```

$$I4 := \frac{\ln(x)x^2}{2} - \int \frac{x}{2} dx$$

The integral  $\int \frac{x}{2} dx$ , is basic so you can apply **value**.

```
> I4 := value(I4);
```

$$I4 := \frac{\ln(x)x^2}{2} - \frac{x^2}{4}$$

Conclude that  $\int x \ln(x) dx = \frac{\ln(x)x^2}{2} - \frac{x^2}{4} + C$ .

Be careful when applying **intparts** that you stick to the principles in the box above, because Maple V can integrate so well that it can evaluate some integrals that have not been reduced to one of the basic formulas. For example, suppose we let  $u = x$  instead of  $u = \ln x$  in the last problem.

```
> I4 := student[intparts](I2, x);
```

$$I4 := x(x \ln(x) - x) - \int x \ln(x) - x dx$$

Now most of us can't directly integrate the above expression, but Maple V can.

```
> value(");
```

$$x(x \ln(x) - x) - \frac{\ln(x)x^2}{2} + \frac{3x^2}{4}$$

```
> simplify(");
```

$$\frac{\ln(x)x^2}{2} - \frac{x^2}{4}$$

This gives the same answer as before. However, if you solve the problem like this, it means that you have missed the point.

Remember that you should regard **intparts** as a tool to help you learn how to integrate by parts. In practical situations you should be able to evaluate easy integrals by hand and compute complicated integrals via Maple V by using the **int** procedure.

Sometimes you might need to apply integration by parts more than once.

**Example 7.3.3** Evaluate  $\int x^2 \sin x \, dx$ .

**Solution:** Using the principles that are suggested for the process of computing an integral by integrating by parts let  $u = x^2$ , and  $v' = \sin x$  for this problem.

```
> I5 := Int(x^2*sin(x), x);
```

$$I5 := \int x^2 \sin(x) dx$$

```
> I6 := student[intparts](I5, x^2);
```

$$I6 := -x^2 \cos(x) - \int -2x \cos(x) dx$$

You are not done yet. The integral on the right hand side has to be integrated by parts with  $u = x$  and  $v' = \cos x$ .

```
> I7 := student[intparts](I6, x);
```

$$I7 := -x^2 \cos(x) + 2x \sin(x) + \int -2 \sin(x) dx$$

The problem has now been reduced to a problem that is easily integrated.

```
> I8 := value(I7);
```

$$I8 := -x^2 \cos(x) + 2x \sin(x) + 2 \cos(x)$$

Conclude that

$$\int x^2 \sin x \, dx = -x^2 \cos(x) + 2x \sin(x) + 2 \cos(x) + C.$$

Sometimes it is not obvious what to let  $u$  and  $v$  be.

**Example 7.3.4** Evaluate  $\int \arcsin x \, dx$ .

**Solution:** In this case let  $u = \arcsin x$  and  $v' = 1$ .

```
> I9 := Int(arcsin(x), x);
```

$$I9 := \int \arcsin(x) dx$$

```
> I10 := student[intparts](I9, arcsin(x));
```

$$I10 := \arcsin(x)x - \int \frac{x}{\sqrt{1-x^2}} dx$$

Now what do we do about the integral on the right-hand side in this case? This is an example of a problem that can be worked by a substitution of  $U = 1 - x^2$ .

```
> I11 := student[changevar](1-x^2=U, I10, U);
```

$$I11 := \arcsin(x)x - \int -\frac{1}{2\sqrt{U}} dU$$

The last integral is basic so integrate and return to the original variables.

```
> I11 := value(I11);
```

$$I11 := \arcsin(x)x + \sqrt{U}$$

```
> I12 := subs(U=1-x^2, I11);
```

$$I12 := \arcsin(x)x + \sqrt{1-x^2}$$

Conclude that

$$\int \arcsin x \, dx = \arcsin(x)x + \sqrt{1-x^2} + C$$

At other times, when using integration by parts one gets the original integral as an intermediate step.

**Example 7.3.5** Evaluate  $\int e^x \cos x \, dx$ .

**Solution:** Here it really doesn't matter which of the two functions you set  $u$  and  $v'$  equal to originally. The results turn out to be essentially the same. We will set  $u = e^x$  and  $v' = \cos x$ .

```
> I13 := Int(exp(x)*cos(x), x);
```

$$I13 := \int e^x \cos(x) dx$$

```
> I14 := student[intparts](I13, exp(x));
```

$$I14 := e^x \sin(x) - \int e^x \sin(x) dx$$

Here is a situation in which the new integral is not simpler than the original integral, but it also is no more complicated than the original integral either. Continue integrating by parts using  $u = e^x$ .

```
> I15 := student[intparts](I14, exp(x));
```

$$I15 := e^x \sin(x) + e^x \cos(x) + \int -e^x \cos(x) dx$$

Now, if you factor out the  $-1$  from the integrand you see that the new integral is identical to the original integral. You can finish this problem by solving algebraically for the original integral.

```
> isolate(I13=simplify(I15), I13);
```

$$\int e^x \cos(x) dx = \frac{e^x \sin(x)}{2} + \frac{e^x \cos(x)}{2}$$

Conclude that

$$\int e^x \cos(x) dx = \frac{e^x \sin(x)}{2} + \frac{e^x \cos(x)}{2} + C$$

### Exercises 7.3

1. Evaluate each of the following integrals by hand if possible. Use Maple V and the **student** package procedure **intparts** for help if necessary.

(a)  $\int t e^{3t} \, dt$

(b)  $\int x^4 \ln x \, dx$

(c)  $\int x^2 \sin 3x \, dx$

(d)  $\int e^{2x} \sin 3x \, dx$

(e)  $\int \arctan 2x \, dx$

(f)  $\int \sec^3 x \, dx$

2. (a) Use integration by parts or **intparts** to show that for any positive integer  $n$ .

$$\int (\ln x)^n \, dx = x(\ln x)^n - n \int (\ln x)^{n-1} \, dx.$$

- (b) Apply the (reduction) formula obtained in the previous part to calculate by hand the exact value of

$$\int_1^3 (\ln x)^4 dx.$$

- (c) Check your answer to part (b) using Maple V.

## 7.4 Using Maple V in Place of Integral Tables

Most functions do not have elementary antiderivatives. The ones that do are so few in number that they can almost all be looked up in tables of integrals. Some students feel that since most of the integrals that have antiderivatives can be looked up in tables that they need not study how to find integrals by their own devices. However, the problems encountered while using tables include some rather sophisticated algebraic techniques such as long division of polynomials, completing the square, and converting rational functions to partial fractions. The student must also develop an ability to recognize the general class of the function that is being integrated. Indeed since tables of integrals are developed by humans, there are errors in the tables. You also need to develop skills in verifying that the results you get are correct.

Maple V is an excellent tool for finding antiderivatives of functions. For this reason its use can virtually replace the need for using integral tables. Remember that there are bugs in any computer programs, including Maple V. Hence, even when using Maple V, you need to develop skills in verifying that the results that you get are correct.

You probably can use Maple V to integrate most any integral that has an elementary integral or that you could use a table to integrate.

We illustrate with a few examples.

**Example 7.4.1** Evaluate  $\int \sin 12x \sin 7x dx$ .

**Solution:** Without Maple V this would might be a rather challenging integration by parts problem. Can you work it that way? You could also use a table of integrals. The problem is easily worked with Maple V.

```
> Int(sin(12*x)*sin(7*x),x) : "=value(")+C;
```

$$\int \sin(12x) \sin(7x) dx = \frac{\sin(5x)}{10} - \frac{\sin(19x)}{38} + C$$

Now that was easy, but how do we know that the answer is correct? This is an example in which verifying the correctness of the answer is more difficult than the actual integration. In order to show the integral is correct differentiate the right-hand side of the last equation.

```
> diff(rhs("),x);
```

$$\frac{\cos(5x)}{2} - \frac{\cos(19x)}{2}$$

Does this look like the integrand in the original problem? You can verify that it is by using trig identities. To see that the integrand is equal to the preceding Maple V output, use the **combine** command with the **trig** option:

```
> combine(sin(12*x)*sin(7*x),trig);
```

$$\frac{\cos(5x)}{2} - \frac{\cos(19x)}{2}$$

This verifies that the value of the integral is correct.

**Example 7.4.2** Evaluate  $\int x^{10} \cos 7x dx$ .

**Solution:**

```
> Int(x^10*cos(7*x),x) : "=value(")+C;
```

$$\int x^{10} \cos(7x) dx = \frac{x^{10} \sin(7x)}{7} + \frac{10x^9 \cos(7x)}{49} - \frac{90x^8 \sin(7x)}{343} - \frac{720x^7 \cos(7x)}{2401} + \frac{720x^6 \sin(7x)}{2401} +$$

$$\frac{4320x^5 \cos(7x)}{16807} - \frac{21600x^4 \sin(7x)}{117649} - \frac{86400x^3 \cos(7x)}{823543} + \frac{259200x^2 \sin(7x)}{5764801} -$$

$$\frac{518400 \sin(7x)}{282475249} + \frac{518400x \cos(7x)}{40353607} + C$$

This time despite the complicated answer the problem is answer is easily checked.

```
> diff(rhs("), x);
```

$$10x^9 \cos(7x) - 7x^{10} \sin(7x)$$

**Example 7.4.3** Evaluate  $\int \frac{x^2+3x-2}{(x+1)(x+2)^2(x^2+6x+14)} dx$ .

**Solution:** This is the kind of problem that you would have to use partial fractions to work by hand. We can use Maple V to evaluate this this integral immediately, but first we will expand the integrand into partial fractions and then integrate. Then we will use Maple V to get the same answer by integrating the problem directly.

```
> f := x -> (x^2+3*x-2)/((x+1)*(x+2)^2*(x^2+6*x+14));
```

$$f := x \mapsto \frac{x^2 + 3x - 2}{(x+1)(x+2)^2(x^2+6x+14)}$$

Use the **convert** command with the **parfrac** option to expand the expression by partial fractions.

```
> g := convert(f(x), parfrac, x);
```

$$g := -\frac{4}{9x+9} + \frac{2}{3(x+2)^2} + \frac{11}{18x+36} - \frac{16+3x}{18x^2+108x+252}$$

The idea of partial fractions is to reduce the rather complicated rational function into a sum of fractions that are easily integrated. Can you integrate each expression in the preceding sum? Remember that you should be able to evaluate each of the integrals in the above expression by hand.

```
> I1 := int(g, x);
```

$$I1 := -\frac{4 \ln(x+1)}{9} - \frac{2}{3x+6} + \frac{11 \ln(x+2)}{18} - \frac{\ln(x^2+6x+14)}{12} - \frac{7\sqrt{5} \arctan\left(\frac{(2x+6)\sqrt{5}}{10}\right)}{90}$$

Now check the result.

```
> diff(I1, x);
```

$$-\frac{4}{9x+9} + \frac{2}{3(x+2)^2} + \frac{11}{18x+36} - \frac{2x+6}{12x^2+72x+168} - \frac{7}{90 + \frac{9(2x+6)^2}{2}}$$

Upon simplifying by using **normal** we arrive at the integrand.

```
> normal(");
```

$$\frac{x^2 + 3x - 2}{(x+1)(x+2)^2(x^2+6x+14)}$$

If all one wants is the answer then one can work the problem in a single step using Maple V.

```
> Int(f(x), x):" = value(") + C;
```

$$\int \frac{x^2 + 3x - 2}{(x+1)(x+2)^2(x^2 + 6x + 14)} dx =$$

$$-\frac{4 \ln(x+1)}{9} - \frac{2}{3x+6} + \frac{11 \ln(x+2)}{18} - \frac{\ln(x^2 + 6x + 14)}{12} - \frac{7\sqrt{5} \arctan(\frac{(2x+6)\sqrt{5}}{10})}{90} + C$$

Sometimes Maple V gives an answer that isn't acceptable, and you can remedy this with the **assume** command.

**Example 7.4.4** Verify the following formula which comes from integral tables.

$$\int \frac{1}{\sqrt{a^2 - x^2}} = \arcsin \frac{x}{a} + C, \quad a \neq 0.$$

**Solution:** In this case, if you do the obvious, you get a rather strange answer, which is unacceptable.

```
> Int(1/sqrt(a^2-x^2), x): "=value(") +C;
```

$$\int \frac{1}{\sqrt{a^2 - x^2}} dx = -I \ln(\sqrt{I}x + \sqrt{a^2 - x^2}) + C$$

What went wrong here? Maple V often gives answers that make little sense to us whenever the problem involves square roots of numbers which may or may not be negative such as, in this case,  $a^2 - x^2$ . If you inform Maple V that you want to assume that  $a > 0$ , then use the **assume** command.

```
> assume(a>0);
```

Now try the same command as before.

```
> Int(1/sqrt(a^2-x^2), x): "=value(") +C;
```

$$\int \frac{1}{\sqrt{a^2 - x^2}} dx = \arcsin\left(\frac{x}{a}\right) + C$$

This time value of the integral is the same one as the one given in tables.

**Exercises 7.4** Evaluate the following integrals. Check your answers though differentiation and simplification.

1.  $\int x^5 \ln x \, dx$
2.  $\int (x^3 - 3x + 5) * e^{3x} \, dx$
3.  $\int \sin^2 t \, dt$
4.  $\int \sin^6 t \, dt$
5.  $\int \frac{3}{9+x^2} \, dx$
6.  $\int \sqrt{(4-x^2)^3} \, dx$
7.  $\int \frac{x^3-2x^2+5}{x^3(x-4)(x^2+2x+2)} \, dx$

## 7.5 Approximating Definite Integrals Numerically

So far, in this chapter, we have studied how to get exact answers of integrals in a number of special cases. However, for most functions it is impossible to find a suitable closed form for an antiderivative, even though we know one exists from the Fundamental Theorem of Calculus.

In Chapters 3 and 6 we discussed the definition of the Riemann Integral and how to obtain bounds on the error encountered in estimating its value with a finite left-hand or right-hand sum.



When the definition of the Riemann Integral of a function  $f$  defined on an interval  $[a, b]$  was given only partitions of the interval that consist on  $n$  equally spaced subintervals of the form

$$a = t_0 < t_1 < \cdots < t_{n-1} < t_n = b,$$

having width

$$\Delta t = t_1 - t_0 = t_2 - t_1 = \cdots = t_n - t_{n-1} = \frac{b - a}{n}$$

were considered. Examples of Riemann Sums were left-hand sums,

$$\text{leftsum} = f(t_0)\Delta t + f(t_1)\Delta t + \cdots + f(t_{n-1})\Delta t = \sum_{k=0}^{n-1} f(t_k)\Delta t,$$

and right-hand sums,

$$\text{rightsum} = f(t_1)\Delta t + f(t_2)\Delta t + \cdots + f(t_n)\Delta t = \sum_{k=1}^n f(t_k)\Delta t.$$

As a matter of fact, the general definition of Riemann Sum uses partitions in which the subintervals are not restricted to be of equal length and the function  $f$  can be evaluated at arbitrary points within each subinterval. For example, if for each  $i$ ,  $i = 1, \dots, n$ , a finite sequence  $z_i$  of points is chosen arbitrarily, except that each  $i$ ,  $z_i$  is taken from the  $i$ th subinterval, *i.e.*, chosen so that  $x_{i-1} \leq z_i \leq x_i$ , then a sum of the form

$$\sum_{i=1}^n f(z_i)\Delta x_i, \quad \text{which} \quad \Delta x_i = x_i - x_{i-1}$$

is also called a Riemann Sum. Another type of Riemann Sum is obtained by selecting  $z_i$  to be the midpoint of the subinterval in which it lies. Sums of this type are called midpoint sums. Recall that the Maple V **student** package has procedures **leftsum** and **rightsum** for obtaining values for left-hand and right-hand sums. The same package also has corresponding graphics procedures, **leftbox** and **rightbox**, that show the figures associated with each type of sum. The **student** package also has the analogous procedures, **middlesum** and **middlebox** for midpoint sums.

An illustration of the use of these procedures is given in the following Maple V segment for

$$f(x) = x(x-1)(x-2) + 1,$$

defined on interval  $[0.5, 2]$ .

```
> f := t -> t*(t-1)*(t-2)+1;
```

$$f := t \mapsto t(t-1)(t-2) + 1$$

We now obtain the left-hand sum accurate to 10 digits for a partition with 10 subintervals.

```
> LeftHandSum := evalf(student[leftsum](f(t), t=0.5..2, 10));
```

```
LeftHandSum := 1.391718750
```

This number represents the sum of the areas of the rectangles shown in Figure 11.

```
> student[leftbox](f(t), t=0.5..2, 10);
```

We now do the same thing with **rightsum** and **rightbox**. See Figure 12.

```
> RightHandSum := evalf(student[rightsum](f(t), t=0.5..2, 10));
```

```
RightHandSum := 1.335468750
```

```
> student[rightbox](f(t), t=0.5..2, 10);
```




Figure 11: Left-hand sum




Figure 12: Right-hand sum

Below is the computation and plot commands necessary for using **middlesum** and **middlebox**. See Figure 13.

```
> MiddleSum := evalf(student[middlesum](f(t), t=0.5..2, 10));

MiddleSum := 1.357265625

> student[middlebox](f(t), t=0.5..2, 10);
```




Figure 13: Midpoint sum

Each of these sums represent an approximation to the definite integral

$$\int_{0.5}^2 t(t-1)(t-2) + 1 \, dt,$$

given to 10 digits of accuracy by

```
> evalf(int(f(t), t=0.5..2));
```

```
1.359375000
```

So which of the Riemann Sums represent the best approximation?

```
> I1-MiddleSum; I1-LeftHandSum; I1-RightHandSum;
```

```
.002109375
```

```
-.032343750
```

```
.023906250
```

Thus we conclude that in this case the procedure **middlesum** gives the numerical value which is closest to the true value of the derivative.

### The Trapezoid Rule

Recall that in Chapter 3, when the problem was to determine how large  $n$  must be in order to approximate the integral to within a prescribed accuracy, we were able to reduce the error by half (for monotone functions) by taking the average of the left-hand sum and the right-hand sum for the same partition. Up to now the numerical integration methods that have been used consisted of certain well defined Riemann Sums. What about a procedure that takes the average of the left-hand sums and the right-hand sums? Continuing with the same illustration above with

$$f(t) = t(t-1)(t-2) + 1,$$

and interval  $[0.5, 2]$  we take the average of the left-hand and right-hand sums.

```
> Average := (LeftHandSum+RightHandSum)/2;
Average := 1.363593750
```

```
> I1 - Average;
```

```
-.004218750
```

At least for this illustration the value of the approximation obtained by taking the average gives a much better approximation than that given by any of the three Riemann Sums above.

It is illuminating to ask about the geometry associated with this averaging method. First let's look at the average algebraically. For the partition

$$a = t_0 < t_1 < \cdots < t_{n-1} < t_n = b$$

with

$$\Delta t = t_1 - t_0 = t_2 - t_1 = \cdots = t_n - t_{n-1} = \frac{b-a}{n},$$

the average of the left-hand and right-hand sums is

$$average = \frac{(leftsum + rightsum)}{2} = \frac{\sum_{k=0}^{n-1} f(t_k) \Delta t + \sum_{k=1}^n f(t_k) \Delta t}{2}$$

We can rearrange the terms of the above to the following form:

$$average = \left( \frac{f(t_0) + f(t_1)}{2} \cdot \Delta t \right) + \left( \frac{f(t_1) + f(t_2)}{2} \cdot \Delta t \right) + \cdots + \left( \frac{f(t_{n-1}) + f(t_n)}{2} \cdot \Delta t \right).$$

How can we interpret this sum? Each expression enclosed by parentheses represents the area of a certain trapezoid. For example, the expression  $\frac{f(t_1) + f(t_2)}{2} \cdot \Delta t$  is the area of the trapezoid with vertices  $(t_1, 0)$ ,  $(t_2, 0)$ ,  $(t_2, f(t_2))$  and  $(t_1, f(t_1))$ . The **student** package has a procedure called **trapezoid** which can calculate this sum.

```
> evalf(student[trapezoid](f(t), t=0.5..2, 10));
```

```
1.363593750
```




Figure 14: Trapezoid sum

Unlike the other three approximation procedures Maple V does not seem to have a corresponding procedure to illustrate the trapezoids. The following Maple V segment plots the 10 trapezoids which occur in the preceding sum along with the graph of  $f(t)$ . See Figure 14.

```
> Plt1 := plots[display]([seq(plots[polygonplot]([[0.5+i*3/20,0],
>   [0.5+(i+1)*3/20,0],[0.5+(i+1)*3/20,f(0.5+(i+1)*3/20)],
>   [0.5+i*3/20,f(0.5+i*3/20)]],i=0..9))):
> Plt2 := plot(f(t),t=0.5..2):
> plots[display]({Plt1,Plt2});
```

### When an Approximation is an Over- or Underestimate





In Chapter 3 it was seen that a left-hand sum taken over an interval in which the function is increasing gives an underestimate and the right-hand sum gives an overestimate. There is a similar statement for a decreasing function. See Figures 15-18.

If  $f$  is increasing on  $[a, b]$ ,

$$\text{leftsum}(f(x), x = a..b, n) \leq \int_a^b f(x) dx \leq \text{rightsum}(f(x), x = a..b, n).$$

If  $f$  is decreasing on  $[a, b]$ ,

$$\text{rightsum}(f(x), x = a..b, n) \leq \int_a^b f(x) dx \leq \text{leftsum}(f(x), x = a..b, n).$$

Figure 15: Left-hand sum  $\leq \int_a^b f(t) dt$ Figure 16:  $\int_a^b f(t) dt \leq$  Right-hand sumFigure 17: Left-hand sum  $\geq \int_a^b f(t) dt$ Figure 18:  $\int_a^b f(t) dt \geq$  Right-hand sum

Suppose that a function  $f$  is concave down over an interval  $[a, b]$ , for example, suppose that  $f'(x) < 0$  on that interval, then  $f$  lies above the line segment joining the points  $(a, f(a))$  and  $(b, f(b))$ . Similarly, a function which is concave up lies below such a line segment. See Figures 19 and 20.



If  $f$  is concave down on  $[a, b]$ ,

$$\text{trapezoid}(f(x), x = a..b, n) \leq \int_a^b f(x) dx.$$

If  $f$  is concave up on  $[a, b]$ ,

$$\int_a^b f(x) dx \leq \text{trapezoid}(f(x), x = a..b, n).$$

The information in the latter box allows one to say something about upper- and lower-estimates for midpoint sum estimates. Let a curve be concave down and take a rectangle whose top intersects the curve at the midpoint

Figure 19: Trapezoid sum  $\leq \int_a^b f(t) dt$ Figure 20:  $\int_a^b f(t) dt \geq$  Trapezoid sum



of the interval. Now consider a tangent to the curve at the the midpoint and consider the trapezoid that is formed. See Figure 21. The trapezoid has the same area as the area of midpoint rectangle, since the triangles formed at the midpoint are congruent. It may be concluded that the midpoint sum overestimates in this case since the upper edge of the trapezoid is above the curve. Similarly, one may conclude that the midpoint sum underestimates the integral when the curve is concave up. The following summarizes these conclusions.

If  $f$  is concave down on  $[a, b]$ ,

$$\text{trapezoid}(f(x), x = a..b, n) \leq \int_a^b f(x) dx \leq \text{middlesum}(f(x), x = a..b, n).$$

If  $f$  is concave up on  $[a, b]$ ,

$$\text{middlesum}(f(x), x = a..b, n) \leq \int_a^b f(x) dx \leq \text{trapezoid}(f(x), x = a..b, n).$$

Figure 21: Midpoint area  $\geq \int_a^b f(t) dt$ Figure 22: Midpoint area  $\leq \int_a^b f(t) dt$

**Example 7.5.1**

For the integral

$$\int_0^{\sqrt{\pi}} \sin t^2 dt$$

perform the following tasks.

1. Find the subinterval of the interval  $[0, \sqrt{\pi}]$  in which the trapezoid rule gives an underestimate and the subinterval in which the trapezoid rule gives an overestimate of the integral.
2. Do the same thing as in (a) except use the midpoint rule.
3. Finally use partitions with 4 subintervals in each of the two subintervals found above and use the appropriate Maple V procedure to obtain under- and overestimates of the value of the integral

**Solution:** According to the statement in the box above we need to determine subintervals of  $[0, \pi]$  in which the function

$$f(t) = \sin t^2$$

is concave up or down.

```
> f := t->sin(t^2);
```

$$f := t \mapsto \sin(t^2)$$

To get an idea of where these intervals are we make a Maple V plot of the function and its second derivative. See Figure 23




Figure 23:  $f(t)$  and its second derivative

```
> plot({f, (D@@2)(f)}, 0..sqrt(Pi));
```

By observing Figure 23 we see that the second derivative changes sign from positive to negative around 0.8. Using **fsolve** we find this point to 10 digits of accuracy.

```
> fsolve((D@@2)(f)(x)=0, x, 0.5..0.9);
```

```
.8082519329
```

We conclude that  $f$  is concave up on the interval

$$[0, 0.8082519329],$$

and concave down on

$$[0.8082519329, \sqrt{\pi}].$$

This means that any trapezoid sum is an underestimate and any midpoint sum is an overestimate of

$$\int_0^{0.8082519329} \sin t^2 dt$$

when taken over the interval  $[0, 0.8082519329]$ . The reverse is true over the interval  $[0.8082519329, \sqrt{\pi}]$ .

```
> Over1 := evalf(student[trapezoid](f(t),t=0..0.8082519329,10));

Over1 := .1714091423

> Under1 := evalf(student[middlesum](f(t),t=0..0.8082519329,10));

Under1 := .1703598876

> Under2 := evalf(student[trapezoid](f(t),t=0.8082519329..sqrt(Pi),10));

Under2 := .7203743485

> Over2 := evalf(student[middlesum](f(t),t=0.8082519329..sqrt(Pi),10));

Over2 := .7259977915

> Underall := Under1+Under2;

Underall := .8907342361

> Overall := Over1+Over2;

Overall := .8974069338

> evalf(int(f(t),t=0..sqrt(Pi)));

.8948314690
```

Thus .8907342361 is an underestimate and .8974069338 is an overestimate for the value the integral

$$\int_0^{\sqrt{\pi}} \sin t^2 dt$$

which to 10 places of accuracy turns out to be 0.8948314690.

### Exercises 7.5

1. Consider the integrals:

$$\int_1^{20} \ln x dx \quad \int_0^5 e^x dx.$$



- (a) For each integral find the left-hand sum, the right-hand sum, and the trapezoid sum for a partition of the appropriate interval with 50 subintervals. Also obtain values for the integrals to 10 digits of accuracy.
- (b) For each integral arrange the left-hand sum, the right-hand sum, and the trapezoid sum and the true value in ascending order. Explain, using Maple V segments and plots where necessary, how you could predict this ordering without doing any calculations in part (a).

2. Consider the integrals

$$\int_{-.7}^{.7} \exp(-x^2) dx \quad \int_{0.8}^1 \exp(-x^2) dx.$$

- (a) For each integral find the trapezoid sum, and the midpoint sum, for a partition of the appropriate interval with 50 subintervals. Also obtain values for the integrals to 10 digits of accuracy.
- (b) For each integral arrange the trapezoid sum, the midpoint sum, and the true value in ascending order. Explain, using Maple V segments and plots where necessary, how you could predict this ordering without doing any calculations in part (a).

## 7.6 Approximation Errors and Simpson's Rule

Any time that you make a numerical approximation you should keep in mind that there is going to be some numerical error. In this section numerical experiments are performed to gain insight into how the error decreases as  $n$  increases for the various approximation methods that have been introduced and, in addition, a more efficient procedure known as Simpson's Rule is presented.

It is known that

$$\int_1^2 \ln x dx = \ln(2) \approx .6931471806.$$

In order to get an idea of how the error reduces as  $n$  increases for the approximation method being used, this known value will be used as a key.

We start our experiment with left- and right-hand sums. Since the **student** package will be used throughout this illustration, the package is made available.

```
> with(student):
```

Define the test function which will be used throughout.

```
> f := x -> 1/x;
```

$$f := x \mapsto x^{-1}$$

The pattern for this experiment is as follows:

1. Apply an approximation rule, such as the left-hand rule, to the function  $f(t)$  over the interval  $[1, 2]$  for  $n = 10, 100, 1000$ , and  $10,000$  subintervals, respectively.
2. Compute the error as the difference between the approximation for the particular value of  $n$  and  $\ln 2$ .
3. Compute the ratio of the error for the previous step and the error at the current step.

For example, if  $n = 10$  the error in the approximating the integral  $\int_1^2 \frac{dt}{t}$  by using **leftsum** is obtained from:

```
> LS1 := evalf(leftsum(f(t), t=1..2, 10));
```

```
LS1 := .7187714032
```

```
> error1 := LS1 - evalf(ln(2));
```

```
error1 := .0256242226
```

Next the error when  $n = 100$  is calculated:

```
> LS2 := evalf(leftsum(f(t),t=1..2,100));
      LS2 := .6956534305

> error2 := LS2 - evalf(ln(2));
      error2 := .0025062499
```

Finally, compute the ratio of the errors.

```
> RL := error1/error2;

      RL := 10.22412913
```

In this case a tenfold increase in  $n$  amounts to an error which is about  $\frac{1}{10}$  of the first error, i.e. an improvement of about one decimal place. Now using this idea one can produce a Maple V segment that will make these computations for  $n = 10, 100, 1000$ , and  $n = 10,000$ .

This segment uses a **do** “loop”. The variable  $LS1$  is assigned the result of calculating the left-hand sum for  $n = 10$ . In the next step one must make sure that the index  $k$  is unassigned so that it can be used as an index in the **do** statement. After that, the loop statement is created. Within the loop the left-hand sum is calculated for  $n = 10^2 = 100$ , and assigned to  $LS2$ . In the next two steps the errors are computed and assigned the values  $error1$  and  $error2$ . Then the ratio of the errors are calculated and assigned to  $Lratio$ . The next step produces a list with four elements that consists of the current value of  $n$ , the old approximation, the current approximation, and the ratio of the two errors. The output is suppressed to save space and the results are summarized in Table 1.

```
> LS1 := evalf(leftsum(f(t),t=1..2,10));
> k := 'k':
> for k from 2 to 4 do
>   LS2 := evalf(leftsum(f(t),t=1..2,10^k));
>   error1 := LS1 - evalf(ln(2));
>   error2 := LS2 - evalf(ln(2));
>   Lratio := error1/error2;
>   [10^k,LS1,LS2,Lratio];
>   LS1 := LS2;
> od;
```

n	Sum for n/10	Sum for n	Ratio of errors
100	.7187714032	.6956534305	10.22412913
1000	.6956534305	.6933972431	10.02249398
10000	.6933972431	.6931721812	10.00225995

Table 1: Table showing ratio of errors as  $n$  increases for left-hand sums

What does the data in Table 1 imply? Notice that if  $n$  is increased by a factor of 10 a decrease in error by a factor of about 10 occurs. It is recommended that you make your own experiments using different values of  $n$ . You will find that the number 10 is not special in that if you increase the number  $n$  by a factor of  $\rho$  then the error will decrease by nearly the same factor. You should also take different functions and repeat the experiment. The same results concerning the ratio by which the error decreases should be nearly the same. The experiment is repeated using right-hand sums.

The following Maple V segment does for right-sums what the previous one did for left-hand sums. Again we do not give the Maple V output but we summarize Table 2.

```
> RS1 := evalf(rightsum(f(t),t=1..2,10));
```

```

> k := 'k':

> for k from 2 to 4 do
> RS2 := evalf(rightsum(f(t),t=1..2,10^k)):
> error1 := RS1 - evalf(ln(2));
> error2 := RS2 - evalf(ln(2));
> Rratio := error1/error2;
> [10^k,RS1,RS2,Rratio];
> RS1 := RS2;
> od;

```

n	Sum for n/10	Sum for n	Ratio of errors
100	.6687714032	.6906534305	9.774747438
1000	.6906534305	.6928972431	9.977494774
10000	.6928972431	.6931221812	9.997739946

Table 2: Table showing ratio of errors as  $n$  increases for right-hand sums

The results in this case are of the same order of magnitude as in the previous one. Observe that a tenfold increase in results in about that much decrease in the error. In this case the ratio is just less than 10. Notice that the ratio seems to be getting closer 10 as  $n$  increases.

The experiment with the trapezoid rule and the midpoint rule suggests that these methods are more efficient. First the midpoint rule. In this case the error became so small for large  $n$  that division became rather inaccurate using the default value of 10 digits. Hence we set **Digits** to 15.

```

> Digits := 15;

Digits := 15

> MS1 := evalf(middlesum(f(t),t=1..2,10)):

> k := 'k':

> for k from 2 to 4 do
> MS2 := evalf(middlesum(f(t),t=1..2,10^k)):
> error1 := MS1 - evalf(ln(2));
> error2 := MS2 - evalf(ln(2));
> Rratio := error1/error2;
> [10^k,MS1,MS2,Rratio];
> MS1 := MS2;
> od;

```

The summary in this case is given in Table 3

If you felt that the midpoint rule gave better results in previous examples then your intuition was correct. In this case when the number of subintervals in the partition is increased by a factor of 10 then the error is reduced by a factor of about 100. In other words, about 2 digits of accuracy are added each time the number of subintervals is increased by a factor of 10.

Next for the trapezoid rule. We leave the **Digits** equal to 16.

n	Sum for n/10	Sum for n	Ratio of errors
100	.692835360409960	.693144055628301	99.7846306762286
1000	.693144055628301	.693147149309952	99.9978350075150
10000	.693147149309952	.693147180247445	99.9999776000000

Table 3: Table showing ratio of errors as n increases for midpoint sums

```

> TS1 := evalf(trapezoid(f(t),t=1..2,10)):
> k := 'k':
> for k from 2 to 4 do
> TS2 := evalf(trapezoid(f(t),t=1..2,10^k)):
> error1 := TS1 - evalf(ln(2));
> error2 := TS2 - evalf(ln(2));
> Rratio := error1/error2;
> [10^k,TS1,TS2,Rratio];
> TS1 := TS2;
> od;

```

The results are summarized in Table 4.

n	Sum for n/10	Sum for n	Ratio of errors
100	.693771403175428	.693153430481824	99.8768668741947
1000	.693153430481824	.693147243059938	99.9987612638613
10000	.693147243059938	.693147181184945	99.9999888000000

Table 4: Table showing ratio of errors as n increases for trapezoid sums

As with the midpoint rule an increase of a factor on 10 in the number of subintervals leads to a corresponding decrease in the error by a factor of 100.

Another rule is defined as a linear combination of the trapezoid rule and the midpoint rule. It is convenient to use Maple V notation in defining this rule:

$$\text{Simpson}(f(t), t = a..b, n) = \frac{2 \cdot \text{middlesum}(f(t), t = a..b, n) + \text{trapezoid}(f(t), t = a..b, n)}{3}.$$

This formula is called Simpson's Rule and a Maple V procedure called **simpson** is contained in the **student** package. In fact the following example shows that the above formula agrees with **simpson**.

```

> evalf(simpson(f(t),t=1..2,20) - (2*middlesum(f(t),t=1..2,10) +
> trapezoid(f(t),t=1..2,10))/3);

```

0

Notice that in the last segment the Maple V procedure **simpson** has  $n = 20$  while for **middlesum** and **trapezoid** was set to  $n = 10$ . A restriction in using the procedure **simpson** is that it requires that the argument  $n$  be even. This is natural from its definition in terms of the midpoint rule and the trapezoid rule using the same partition with, say,  $m$  subintervals. The values for the trapezoid rule are taken at the endpoints of the subintervals, while the values for the midpoint rule are evaluated along the midpoints of the subintervals. Thus, for all practical purposes, evaluating both of these sums for a partition with  $m$  subintervals is equivalent to having a partition with  $2m$  subintervals.

We now evaluate **simpson** for values of  $n = 20, 200, 2000$ , and  $20,000$ . We find that we need to set **Digits** = 20 to keep from dividing by zero for the larger values of  $n$  in this case.

```

> Digits := 20;

                               Digits := 20

> SS1 := evalf(simpson(f(t),t=1..2,2*10)):
> k := 'k':
> for k from 2 to 4 do
> SS2 := evalf(simpson(f(t),t=1..2,2*10^k)):
> error1 := SS1 - evalf(ln(2));
> error2 := SS2 - evalf(ln(2));
> Rratio := error1/error2;
> [10^k,SS1,SS2,Rratio];
> SS1 := SS2;
> od;

```

Table 5 gives a summary. Note that an increase for  $n$  by a factor of 10 yields a corresponding decrease in the error by a factor of nearly  $10^4 = 10,000$ . This means with each increase in  $n$  by a factor of 10 one makes a gain of around 4 digits of accuracy.

n	Sum for n/10	Sum for n	Ratio of errors
100	.69314737466511611897	.69314718057947533885	9938.8058530718762977
1000	.69314718057947533885	.69314718055994726254	9999.4006666257065618
10000	.69314718055994726254	.69314718055994530961	10279.578947368421053

Table 5: Table showing ratio of errors as  $n$  increases for Simpson's Rule

Simpson's Rule is a very important method for numerical integration and provides a reasonable degree of accuracy for modestly small values of  $n$ .

In practice you use something like Simpson's Rule when you don't already know the answer. When do you know when you have the accuracy that you want to have? Suppose that you wish to be sure that the answer is correct to 4 decimal places. One rather heuristic procedure is to keep increasing  $n$  until the first four decimal places do not change with successive estimates. The following example illustrates one such example.

**Example 7.6.1** Use Simpson's Rule to estimate the value

$$\int_0^{\sqrt{\pi}} \sin x^2 dx$$

with an error less than  $10^{-4}$ .

**Solution:** We will use a **for** loop and compute approximations using Simpson's Rule for values of  $2 \cdot n$  for  $n$  from 5 to 10. Looking at the values we will determine (heuristically) a value for the integral accurate to four decimal places.

```

> for n from 5 to 10 do

> [n,evalf(simpson(sin(t^2),t=0..sqrt(Pi),2*n))];

> od;

[5, .8950818902]

```

```
[6, .8949514362]
[7, .8948959374]
[8, .8948691440]
[9, .8948549391]
[10, .8948468435]
```

Observe that from  $n = 7$  through  $n = 10$  the first four decimal places does not change. The desired accuracy seems to have been obtained with  $n = 7$ . We now double this value and check the approximation with  $n = 14$ .

```
> evalf(simpson(sin(t^2),t=0..sqrt(Pi),2*14));
.8948354575
```

Since the approximation for  $n = 7$  and 14 differ by less than .0000604799, we will consider .8948 as an acceptable answer for this approximation.

### Exercises 7.6

- For the following problems use Simpson's Rule with various values of  $n$  to evaluate the definite integrals with an error less than 0.0001. Explain why you believe that you have a valid approximation.

(a)  $\int_0^2 \frac{dx}{9+x^2}$

(b)  $\int_0^1 \sin(\sin(x)) dx$

(c)  $\int_1^2 \sin(x^2 + 3x + 1) dx$

- From the fact that

$$\frac{d}{dx}(\arctan)(x) = \frac{1}{1+x^2}$$

we know that

$$\int \frac{dx}{1+x^2} = \arctan(x) + C.$$

Thus we have the following formula for computing the number  $\pi$  :

$$\pi = 4 \int_0^1 \frac{dx}{1+x^2}.$$

Use this formula and Simpson's Rule to estimate  $\pi$  to an accuracy with error less than 0.00001. Explain why you believe that you have a valid approximation.

## 7.7 Improper Integrals

Up to now all of the functions that have been studied with regard to integration have been bounded and defined on intervals of finite length. In this section you will learn how to deal with integrals like the following:

$$\int_0^\infty e^{-x^2} dx$$

and

$$\int_0^1 \frac{dx}{\sqrt{x(1-x)}}.$$

Integrals like the last two are called improper integrals.

For example, consider the integral

$$\int_0^T e^{-x^2} dx,$$

for various values of  $T > 0$ .

```
> evalf(int(exp(-x^2), x=0..2));
.8820813910

> evalf(int(exp(-x^2), x=0..4));
.8862269120

> evalf(int(exp(-x^2), x=0..5));
.8862269255

> evalf(int(exp(-x^2), x=0..6));
.8862269255

> evalf(int(exp(-x^2), x=0..7));
.8862269255
```

Observe, that the value of the integral for  $T = 5$  through  $T = 7$  are unchanged to 10 digits of accuracy. We know that  $e^{-x^2} > 0$  for all  $x$ . Thus we know that the area under the curve  $y = e^{-x^2}$  between  $x = 5$  and  $x = 7$  must be positive. We also know that

$$\int_0^7 e^{-x^2} dx = \int_0^5 e^{-x^2} dx + \int_5^7 e^{-x^2} dx$$

and so

$$\int_0^5 e^{-x^2} dx < \int_0^7 e^{-x^2} dx.$$

Looking at the situation from a geometric point of view we now plot the indefinite integral of  $e^{-x^2}$  over the interval  $[0, 7]$ .

```
> plot(int(exp(-t^2), t=0..x), x=0..7);
```

According Figure 24 the graph of the antiderivative

$$F(x) = \int_0^x e^{-t^2} dt$$

appears to have a horizontal asymptote. Can we prove this? We know that  $F'(x) = e^{-x^2} > 0$  for all  $x$ . Thus we know that  $F(x)$  is monotone increasing function. Since  $F(x)$  is increasing we can show that


$$\lim_{x \rightarrow \infty} F(x)$$

exists if  $F(x)$  is bounded above. Can we show that  $F(x)$  is bounded? Well for one thing we know that for  $x \geq 1$  that  $x^2 \geq x$  and hence

$$e^{-x^2} \leq e^{-x}, \quad \text{for } x \geq 1.$$

Now consider the function  $G(x)$  defined for  $x \geq 1$  by

$$G(x) = \int_1^x e^{-t} dt = e^{-1} - e^{-x}.$$

Figure 24: Antiderivative of  $e^{-x^2}$  over  $[0, 7]$ 

Now since  $e^{-x^2} \leq e^{-x}$  for  $x \geq 1$ , we know from Chapter 6 that

$$\int_1^x e^{-t^2} dt \leq \int_0^x e^{-t} dt = e^{-1} - e^{-x} \leq \frac{1}{e}.$$

Since  $e^{-x^2}$  is continuous for all  $x$  it is clear that  $\int_0^1 e^{-t^2} dt$  is a finite number. Hence, for all  $x$  the function  $F(x)$  satisfies

$$F(x) \leq \int_0^1 e^{-t^2} dt + \frac{1}{e}.$$

We conclude that the limit

$$\lim_{x \rightarrow \infty} F(x) = \lim_{x \rightarrow \infty} \int_0^x e^{-t^2} dt$$

exists. The following gives a definition for an improper integral on a semi-infinite interval.

**Improper integral over  $[a, \infty)$**

Let  $f$  be defined and integrable over every interval  $[a, T]$  for  $T \geq a$ , then the improper integral

$$\int_a^\infty f(t) dt$$

is defined to be the number given by

$$\lim_{T \rightarrow \infty} \int_a^T f(t) dt,$$

if this limit exists.

In the case that the limit exists we say that the integral *converges*.

If the limit does not exist we say that the integral *diverges*.

According to this definition the improper integral

$$\int_0^\infty e^{-t^2} dt$$



exists and is equal to the number .8862269255 to ten digits of accuracy.

Consider the improper integral

$$\int_1^{\infty} \frac{dt}{t}.$$

This integral is divergent since

$$\int_1^T \frac{1}{t} dt = \ln(T),$$

and

$$\lim_{T \rightarrow \infty} \int_1^T \frac{1}{t} dt = \lim_{T \rightarrow \infty} \ln T = \infty.$$

Now consider the function  $f(t) = \frac{1}{\sqrt{t}}$  which is defined for all  $t > 0$ , but is unbounded in the vicinity of  $t = 0$ . Now for  $\epsilon > 0$  we have

$$\int_{\epsilon}^1 \frac{1}{\sqrt{t}} dt = 2\sqrt{t} \Big|_{\epsilon}^1 = 2 - 2\sqrt{\epsilon}.$$

Thus

$$\lim_{\epsilon \rightarrow 0} 2 - 2\sqrt{\epsilon} = 2.$$

#### Improper integral for Unbounded Integrand

Let  $f$  be defined and integrable over every interval of the form  $[a + \epsilon, b]$  for positive  $\epsilon$ . Suppose that  $f$  is unbounded in the vicinity of  $t = a$ , then the improper integral

$$\int_a^b f(t) dt$$

is defined to be the number given by

$$\lim_{\epsilon \rightarrow 0} \int_{a+\epsilon}^b f(t) dt,$$

if this limit exists.

In the case that the limit exists we say that the integral *converges*.

If the limit does not exist we say that the integral *diverges*.

Therefore, the integral

$$\int_0^1 \frac{1}{\sqrt{t}} dt$$

converges to 2.

**Example 7.7.1** Determine that the integral

$$\int_2^{\infty} \frac{1}{t^2 \ln t} dt$$

converges. Find the limit to 10 digits of accuracy.

**Solution:** Let

$$F(x) = \int_2^x \frac{1}{t^2 \ln t} dt.$$

Since  $\ln$  is increasing we have  $\ln t \geq \ln 2$  for  $t \geq 2$ . This means that

$$\frac{1}{t^2 \ln t} \leq \frac{\ln 2}{t^2}$$

for all  $t \geq 2$ . It follows that

$$F(x) = \int_2^x \frac{1}{t^2 \ln t} dt \leq \int_2^x \frac{1}{t^2} dt = 1/2 - 1/x \leq 1/2,$$

for all  $x \geq 2$ . Since  $F'(x) = \frac{1}{x^2 \ln x} > 0$  for  $x > 1$ ,  $F(x)$  is an increasing function which is bounded above by  $\frac{1}{2}$ . Thus the integral converges. Its value to 10 digits of accuracy is given by the following.

```
> I1 := Int(1/(x^2*ln(x)), x=2..infinity);
      I1 := \int_2^{\infty} \frac{1}{x^2 \ln(x)} dx
> I1 := evalf(I1);

      I1 := .3786710430
```

**Example 7.7.2** Determine that the integral

$$\int_0^1 \frac{1}{t^2 + \sqrt{t}} dt$$

converges. Find the limit to 10 digits of accuracy.

**Solution:** Let

$$F(x) = \int_x^1 \frac{1}{t^2 + \sqrt{t}} dt$$

for  $0 < x \leq 1$ . Since  $t^2 > 0$  for all  $t \neq 0$ , we have  $\frac{1}{t^2 + \sqrt{t}} < \frac{1}{\sqrt{t}}$ . This means

$$\int_x^1 \frac{1}{t^2 + \sqrt{t}} dt \leq \int_x^1 \frac{1}{\sqrt{t}} dt = 2 - 2\sqrt{x} \leq 2,$$

for all  $0 < x < 1$ . Now  $F'(x) = -\frac{1}{x^2 + \sqrt{x}} < 0$  and  $F(x)$  is decreasing as  $x$  increases. This means that  $F(x)$  increases as  $x$  decreases to 0. Since  $F(x)$  is bounded above by 2, it converges to some limit,

$$\int_0^1 \frac{1}{t^2 + \sqrt{t}} dt,$$

which can not be greater than 2. The following Maple V segment obtains the value of the improper integral to 10 digits of accuracy.

```
> I2 := Int(1/(x^2+sqrt(x)), x=0..1);
      I2 := \int_0^1 (x^2 + \sqrt{x})^{-1} dx
> I2 := evalf(I2);

      I2 := 1.671297697
```

**Exercise 7.7** Determine whether the following integrals are convergent or divergent. Evaluate the integral if it is convergent.

1.  $\int_0^1 \sqrt{\frac{1+x}{1-x}} dx$  (Hint: Compare the integrand with  $2\frac{1}{\sqrt{1-x}}$ .)
2.  $\int_2^{\infty} \frac{1}{\ln^2 t} dt$  (Hint: Compare the integrand with  $\frac{1}{t \ln t}$ .)

3.  $\int_0^{\infty} \frac{1}{t+e^t} dt$  (Hint: Compare the integrand with  $e^{-t}$ .)

4.  $\int_0^3 \frac{dt}{(t-1)^{2/3}}$  (Hint: Consider two improper integrals:  $\int_0^1 \frac{dt}{(t-1)^{2/3}}$  and  $\int_1^3 \frac{dt}{(t-1)^{2/3}}$ .)